_<X>:=PolynomialRing(Rationals());
R<h,x>:=PolynomialRing(Rationals(),2);
S<h,x>:=FunctionField(Rationals(),2);
PS<t>:=PolynomialRing(S);

print "Z2xZ20 Case";
j10:=(h^6-4*h^5+16*h+16)^3*(h+1)^(-2)*(h-4)^(-1)*h^(-5);
//j-invariant for 10-isogeny
E:=EllipticCurve([1,0,0,-36/(j10-1728),-1/(j10-1728)]);
WE:=WeierstrassModel(E);
//Indeed all such curves have a 2-torsion point over Q
D:=DivisionPolynomial(WE,2);
Dfact:=Factorization(D);
g:=x^3+aInvariants(WE)[2]*x^2+aInvariants(WE)[4]*x+aInvariants(WE)[5];
f:=Evaluate(g,[h,t+Roots(Dfact[1][1])[1][1]]);
//moving the 2-torsion point to (0,0)
f:=PS!f;
b:=Coefficients(f)[3];
d:=Coefficients(f)[2];
F:=(b^2-4*d)*d;
//want to know when F is a square, want y^2 = F, can absorb squares
numfact:=Factorization(Numerator(F));
denfact:=Factorization(Denominator(F));
FF:=R!1;
for i:=1 to #numfact do
	if IsOdd(numfact[i][2]) then
		 FF:=FF*R!numfact[i][1];
	end if;
end for;
for i:=1 to #denfact do
	if IsOdd(denfact[i][2]) then
		 FF:=FF*R!denfact[i][1];
	end if;
end for;
FF;
//FF:=h^3 - 3*h^2 - 4*h
E:=EllipticCurve(X^3-3*X^2-4*X);
Rank(E);
MordellWeilGroup(E);
PointSearch(E,1000);
//it turns out y^2 = FF is an elliptic curve with rank 0, torsion subgroup Z/2xZ/2,
//the points having x-coords: h=-1, h=4, h=0, all of which are cusps.
G:=d;
//want to know when G is a square, want y^2 = G, can absorb squares
numfact:=Factorization(Numerator(G));
denfact:=Factorization(Denominator(G));
GG:=R!1;
for i:=1 to #numfact do
	if IsOdd(numfact[i][2]) then
		 GG:=GG*R!numfact[i][1];
	end if;
end for;
for i:=1 to #denfact do
	if IsOdd(denfact[i][2]) then
		 GG:=GG*R!denfact[i][1];
	end if;
end for;
GG;
E:=EllipticCurve(X^3+X^2+4*X+4);
Rank(E);
MordellWeilGroup(E);
PointSearch(E,1000);
//GG:=h^3+h^2+4*h+4
//it turns out y^2 = GG is an elliptic curve with rank 0, torsion subgroup Z/6
//the points having x-coords: h=-1, h=4, h=0, all of which are cusps.


print "Z2xZ36 Case";
j18:=(h^3-2)^3*(h^9-6*h^6-12*h^3-8)^3*h^(-9)*(h^3-8)^(-1)*(h^3+1)^(-2);
//j-invariant for 18-isogeny
E:=EllipticCurve([1,0,0,-36/(j18-1728),-1/(j18-1728)]);
WE:=WeierstrassModel(E);
//Indeed all such curves have a 2-torsion point over Q
D:=DivisionPolynomial(WE,2);
Dfact:=Factorization(D);
g:=x^3+aInvariants(WE)[2]*x^2+aInvariants(WE)[4]*x+aInvariants(WE)[5];
f:=Evaluate(g,[h,t+Roots(Dfact[1][1])[1][1]]);
//moving the 2-torsion point to (0,0)
f:=PS!f;
b:=Coefficients(f)[3];
d:=Coefficients(f)[2];
F:=(b^2-4*d)*d;
//want to know when F is a square, want y^2 = F, can absorb squares
numfact:=Factorization(Numerator(F));
denfact:=Factorization(Denominator(F));
FF:=R!1;
for i:=1 to #numfact do
	if IsOdd(numfact[i][2]) then
		 FF:=FF*R!numfact[i][1];
	end if;
end for;
for i:=1 to #denfact do
	if IsOdd(denfact[i][2]) then
		 FF:=FF*R!denfact[i][1];
	end if;
end for;
FF;
//FF:=h^7 - 7*h^4 - 8*h
//Now we make the hyperelliptic curve y^2 = FF

PP<X,Y,Z>:=ProjectiveSpace(Rationals(),2);
C:=Curve(PP, Y^2*Z^5 - X^7 + 7*X^4*Z^3 + 8*X*Z^6); C;
AutC:=Automorphisms(C);
G:=AutomorphismGroup(C); G; #G;
G.3;
GG:=AutomorphismGroup(C,[SchemeMap(G.3)]); GG;
Cg, prj:=CurveQuotient(GG); Cg; prj;
Degree(prj);

MordellWeilGroup(Cg);
//Shows there are only 4 points
//Need to pull back points to C
Points:=PointSearch(Cg,10000); 
//Strangely it leaves out the points (1:1:1), we append it manually.
Append(~Points, Cg![1,1,1]);

Points;

for P in Points do
    preimageofP:= P @@ prj;
    RationalPoints(preimageofP);
end for;

prj(C![-1,0,1]);
prj(C![0,0,1]);
prj(C![2,0,1]);
prj(C![0,1,0]);


G:=d;
//want to know when G is a square, want y^2 = G, can absorb squares
numfact:=Factorization(Numerator(G));
denfact:=Factorization(Denominator(G));
GG:=R!1;
for i:=1 to #numfact do
	if IsOdd(numfact[i][2]) then
		 GG:=GG*R!numfact[i][1];
	end if;
end for;
for i:=1 to #denfact do
	if IsOdd(denfact[i][2]) then
		 GG:=GG*R!denfact[i][1];
	end if;
end for;
GG;
E:=EllipticCurve(X^3+1);
Rank(E);
MordellWeilGroup(E);
PointSearch(E,1000);
//GG:=h^3+1
//it turns out y^2 = GG is an elliptic curve with rank 0, torsion subgroup Z/6
//the points having x-coords: h=-1, h=2, h=0, all of which are cusps.

